

UKŁADY SEKWENCYJNO–CZASOWE I SPECJALNE

Zbiornik z dwoma zaworami. Reaktor chemiczny. Powtarzanie impulsu o mierzonym czasie trwania. Generacja chwilowego impulsu po zadanym czasie. Zespół trzech zbiorników.


ZBIORNIK Z DWOMA ZAWORAMI

1. Zadanie


Zbiornik jest napełniany zaworem Z1, a opróżniany zaworem Z3. Przebiega to następująco:

- Dla $h < X1$ otwarty jest zawór Z1.
- Jeśli h przekroczy $X1$, to Z1 jest otwarty przez 20 sekund, chyba że wcześniej h osiągnie $X2$ (wtedy Z1 zamyka się natychmiast).
- Po zamknięciu Z1 otwierany jest Z3, aż h spadnie do $X1$.

Dodatkowy zawór Z2 jest otwierany ręcznie.



2. Przebiegi czasowe




3. Automat

- Stany

- ① nalewanie, gdy $h < X1 \rightarrow Z1 = 1, Z3 = 0$
- ② nalewanie, gdy $h > X1$, ale nie dłużej niż przez 10 sekund lub do momentu, gdy h osiągnie $X2$ ($h = X2$) $\rightarrow Z1 = 1, Z3 = 0$
- ③ oczekiwanie na opróżnienie, tzn. gdy $h = X1 \rightarrow Z1 = 0, Z3 = 1$

- Graf



4. Kod

- STEROWNIK

```

STEROWNIK (PRG-ST)
0001 PROGRAM STEROWNIK
0002 VAR_EXTERNAL
0003   x1,x2,Z1,Z3: BOOL;
0004 END_VAR
0005 VAR
0006   stan: INT:=1;
0007   TON_1:TON: T: TIME := T#10s;
0008   _0:BOOL:=FALSE; _1:BOOL:=TRUE;
0009 END_VAR
0001 TON_1(IN:=stan=2, PT:=T);
0002
0003 CASE stan OF
0004 1: Z1:=_1; Z3:=_0;
0005   IF x1 THEN stan:=2; END_IF
0006 2: Z1:=_1; Z3:=_0;
0007   IF NOT x1 THEN stan:=1;
0008   ELSIF x2 OR TON_1.Q THEN stan:=3; END_IF
0009 3: Z1:=_0; Z3:=_1;
0010   IF NOT x1 THEN stan:=1; END_IF
0011 END_CASE
  
```

Kod ściśle odpowiada grafowi automatu.

Stałe `_0`, `_1` reprezentują FALSE i TRUE (kod wygląda czytelniej).

- ZBIORNIK i Obraz_poziomu

Przepływ zakłócający `Q_Z2` przez zawór `Z2` jest ustawiony suwakiem. Oddziaływanie na poziom reprezentuje instrukcja



$$H := H + Q_{Z2};$$

Maksymalna wartość `Q_Z2` wynosi 0.1, tzn. 2-krotnie więcej niż odpływ przez zawór `Z3` (0.05). Dlatego formatem wyświetlacza przy suwaku jest `%.2f`.

```

ZBIORNIK (PRG-ST)
0001 PROGRAM ZBIORNIK
0002 VAR_EXTERNAL
0003   x1,x2,Z1,Z3: BOOL;
0004 END_VAR
0005 VAR
0006   H:REAL; (* poziom *)
0007   Q_Z2:REAL; (* dopływ z Z2 *)
0008 END_VAR
0009
0001 H:=H+Q_Z2;
0002 IF Z1 THEN H:=H+0.05; END_IF
0003 IF Z3 THEN H:=H-0.05; END_IF
0004
0005 IF H<0.0 THEN H:=0.0; END_IF
0006 IF H>100.0 THEN H:=100.0; END_IF
0007
0008 IF H>80.0 THEN x2:=TRUE;
0009 ELSE x2:=FALSE; END_IF
0010 IF H>20.0 THEN x1:=TRUE;
0011 ELSE x1:=FALSE; END_IF
  
```

- STEROWNIK w LD





REAKTOR CHEMICZNY

1. Zadanie

Należy zaprojektować układ sterowania reaktorem funkcjonujący według poniższego algorytmu:

- Nalewanie na zmianę z zaworów Z1/Z2 w cyklu 3+3 s do poziomu X2
- Nalewanie dalej z Z1 aż do X3 (Z2 zamknięty)
- Zamknięcie Z1, włączenie grzejnika GR, aż temperatura osiągnie TM
- Wyłączenie GR i oczekiwanie przez 10 s.
- Opróżnianie z Z3 w cyklu 4+2 s (otw./zamkn.) do X1
- Zamknięcie Z3 i powrót na początek (nalewanie)






2. Automat

- Stany

- ① nalewanie z Z1 przez 3 s, $h < X2$
- ② nalewanie z Z2 przez 3 s, $h < X2$
- ③ nalewanie z Z1 do X3
- ④ grzanie do TM
- ⑤ oczekiwanie przez 10 s
- ⑥ opróżnianie, Z3 otwarty przez 4 s.
- ⑦ Z3 zamknięty przez 2 s.

3. Graf



Uwaga. Ten sam czasomierz można wykorzystać w kilku stanach byleby nie następowały one po kolei.

4. STEROWNIK i REAKTOR

- W typowej realizacji wykorzystywane byłyby pięć czasomierzy, po jednym dla każdego przedziału czasu. Można jednak ograniczyć się tylko do dwóch pod warunkiem, że na wejściach PT przez przejściem do odpowiednich stanów ustawiane zostaną właściwe wartości. Ilustruje to poniższy kod ściśle odpowiadający grafowi.

```

STEROWNIK (PRG-ST)
0001 PROGRAM STEROWNIK
0002
0003 VAR_EXTERNAL X1,X2,X3,TM,Z1,Z2,Z3,GR: BOOL; END_VAR
0004
0005 VAR stan: INT:=1; TON_1:TON; TON_2:TON; T1: TIME := T#3s;
0006 T2: TIME := T#3s; _0:BOOL:=FALSE; _1:BOOL:=TRUE;
0007 END_VAR
0008
0009 TON_1(IN:=stan=1 OR stan=5 OR stan=7, PT:=T1);
0010 TON_2(IN:=stan=2 OR stan=6, PT:=T2);
0011
0012 CASE stan OF
0013 1: Z1:=_1; Z2:=_0; Z3:=_0; GR:=_0;
0014 IF TON_1.Q THEN T2:=T#3s; stan:=2;
0015 ELSIF X2 THEN stan:=3; END_IF
0016 2: Z1:=_0; Z2:=_1; Z3:=_0; GR:=_0;
0017 IF TON_2.Q THEN T1:=T#3s; stan:=1;
0018 ELSIF X2 THEN stan:=3; END_IF
0019 3: Z1:=_1; Z2:=_0; Z3:=_0; GR:=_0;
0020 IF X3 THEN stan:=4; END_IF
0021 4: Z1:=_0; Z2:=_0; Z3:=_0; GR:=_1;
0022 IF TM THEN T1:=T#10s; stan:=5; END_IF
0023 5: Z1:=_0; Z2:=_0; Z3:=_0; GR:=_0;
0024 IF TON_1.Q THEN T2:=T#4s; stan:=6; END_IF
0025 6: Z1:=_0; Z2:=_0; Z3:=_1; GR:=_0;
0026 IF TON_2.Q THEN T1:=T#2s; stan:=7;
0027 ELSIF NOT X1 THEN T1:=T#3s; T2:=T#3s; stan:=1; END_IF
0028 7: Z1:=_0; Z2:=_0; Z3:=_0; GR:=_0;
0029 IF TON_1.Q THEN T2:=T#4s; stan:=6; END_IF
0030 END_CASE

REAKTOR (PRG-ST)
0001 PROGRAM REAKTOR
0002 VAR_EXTERNAL
0003 X1,X2,X3,TM,Z1,Z2,Z3,GR: BOOL;
0004 END_VAR
0005 VAR
0006 H:REAL; TEMP:REAL:=20.0;
0007 END_VAR
0008
0009 IF Z1 THEN H:=H+0.05; END_IF
0010 IF Z2 THEN H:=H+0.05; END_IF
0011 IF Z3 THEN H:=H-0.05; END_IF
0012 IF H<0.0 THEN H:=0.0; END_IF
0013 IF H>100.0 THEN H:=100.0; END_IF
0014
0015 IF H>=80.0 THEN X3:=TRUE;
0016 ELSE X3:=FALSE; END_IF
0017 IF H>=50.0 THEN X2:=TRUE;
0018 ELSE X2:=FALSE; END_IF
0019 IF H>=20.0 THEN X1:=TRUE;
0020 ELSE X1:=FALSE; END_IF
0021
0022 IF Z1 OR Z2 THEN TEMP:=20.0; END_IF
0023 IF GR THEN TEMP:=TEMP+0.06; END_IF
0024
0025 IF TEMP>=80.0 THEN TM:=TRUE;
0026 ELSE TM:=FALSE; END_IF
0027

```

- Temperaturę reprezentuje zmienna TEMP rosnąca o 0.06°C w ciągu 1 cyklu obliczeń, tzn.


$$TEMP := TEMP + 0.06;$$

Temperatura maksymalna TM wynosi 80 °C, zatem wzrost od 20 do 80 °C przy cyklu 10 ms=0.01 s nastąpi w ciągu czasu

$$\frac{(80 - 20)}{0.06} \cdot 0.01 = 10s$$

5. Obraz reaktora

- Spirala grzejna zmienia kolor wraz ze zmienną GR. Bargraf po lewej stronie spirali reprezentuje TEMP.




6. „Nie krócej i nie dłużej”

- Jedną z form pośredniej kontroli prawidłowości pomiaru temperatury podczas grzania jest sprawdzanie, czy grzanie nie trwa zbyt krótko lub zbyt długo (co zdarza się przy przerwaniu lub zwarciu przewodów pomiarowych).

Załóżmy, że trzeci warunek zadania sterowania reaktorem zostanie rozszerzony do postaci:

- Zamknięcie Z1, włączenie grzejnika GR, aż temperatura osiągnie TM, ale *nie krócej niż 5 s i nie dłużej niż 20 s*.
- Realizacja takiego warunku wymaga uzupełnienia powyższego programu o czasomierz TON_3 aktywny w stanie 4, którego wyjście ET (*Elapsed Time*) służy do pomiaru czasu. Wejście PT (*Preset Time*) powinno być wtedy ustawione na wartość przekraczającą 20 s (np. 30 s).
- Przejście między stanami 4 i 5 (zob. graf)



- Uzupełnienia programu STEROWNIK

- Deklaracja czasomierza TON_3

```
TON_3:TON;
```

- Wywołanie czasomierza

```
TON_3(IN:=stan=4, PT:=T#30s);
```

- Obsługa stanu 4

```
4: Z1:=_0; Z2:=_0; Z3:=_0; GR:=_1;
```


```
IF TM AND TON_3.ET>=T#5s AND TON_3.ET<=T#20s
```

```
THEN T1:=T#10s; stan:=5; END_IF
```

POWTARZANIE IMPULSU O MIERZONYM CZASIE TRWANIA


1. Zadanie

Mierzony jest czas trwania T_x impulsu wejściowego, po którym wyjście powtarza ten impuls. Zmiany wejścia w czasie powtarzania są ignorowane.




2. Stany i graf

- ① oczekiwanie na wejście
- ② pomiar czasu T_x
- ③ powtarzanie impulsu na wyjściu (dokładne)
- ④ oczekiwanie na zanik wejścia



Wyjaśnienia. Mierzony czas jest podany na wyjściu ET czasomierza TON, którego wejście PT należy ustawić na wartość większą niż najdłuższy spodziewany impuls (np. $PT:=T\#3600s$). Impuls o zmierzonym czasie trwania jest generowany na wyjściu przez czasomierz TP.


3. Kod i obraz




Uwaga. Wyjście TON.ET (pomiar T_x) jest wyrażone w milisekundach.

4. Przerwanie powtarzania (modyfikacja zadania)

Jeżeli w czasie powtarzania impulsu wejście zostało ustawione ponownie, to powtarzanie jest przerywane i czas T_x mierzony od nowa.



- Graf



Uwaga. Czas T_x odmierzany w stanie 3 jest teraz czasem maksymalnym (poprzednio był odtwarzany dokładnie).


- Kod i obraz

```

MAIN (PRG-ST)
0001 PROGRAM MAIN
0002 VAR
0003   stan: INT := 1;
0004   we, wy: BOOL;
0005   TON1: TON; TP1: TP;
0006   PT: TIME;      (* dotyczy TP1 *)
0007 END_VAR
0008
0001 TON1(IN:=stan=2, PT:=T#3600s);
0002 TP1(IN:=stan=3, PT:=PT);
0003
0004 CASE stan OF
0005 1: wy:=FALSE;
0006   IF we THEN stan:=2; END_IF
0007 2: wy:=FALSE;
0008   IF NOT we THEN PT:=TON1.ET; stan:=3; END_IF
0009 3: wy:=TRUE;
0010   IF NOT TP1.Q THEN stan:=1;
0011   ELSIF we AND TP1.Q THEN stan:=2; END_IF
0012 END_CASE
0013

```

- MAIN w LD




Uwaga. W gałęzi przejścia ze stanu 2 do 3 znajduje się również podstawienie PT:=TON1.ET.

GENERACJA CHWILOWEGO IMPULSU PO ZADANYM CZASIE


1. Zadanie

Impuls trwający 1 cykl pojawia się 1.5 sekundy od ustawienia wejścia. Zmiany wejścia w tym czasie są ignorowane.



2. Stany i graf

- | | |
|--------------------------|--------------------------------|
| ① oczekiwanie na wejście | ③ impuls trwający 1 cykl |
| ② odmierzenie 1.5 s | ④ oczekiwanie na zanik wejścia |



Uwaga. Stan 3 nie zawiera pętli „pozostawania w stanie” bo trwa tylko 1 cykl.

3. Kod i obraz

```


0001 PROGRAM MAIN
0002 VAR
0003     stan: INT := 1; licz: INT := 0;
0004     we, wy: BOOL;
0005     TON1: TON; TP1: TP;
0006 END_VAR
0007
0008 TON1(IN:=stan=2, PT:=T#1s500ms);
0009
0010 CASE stan OF
0011 1: wy:=FALSE;
0012   IF we THEN stan:=2; END_IF
0013 2: wy:=FALSE;
0014   IF TON1.Q THEN stan:=3; END_IF
0015 3: wy:=TRUE; licz:=licz+1;
0016   IF NOT we THEN stan:=1;
0017   ELSE stan:=4; END_IF
0018 4: wy:=FALSE;
0019   IF NOT we THEN stan:=1; END_IF
0020 END_CASE
0021
0022 TP1(IN:=wy, PT:=T#200ms);
  
```

Wyjaśnienia. Generowane impulsy zlicza licznik. LED jest aktywowany wyjściem TP1.Q czasomierza generującego impuls trwający 200 ms, ponieważ właściwego impulsu wy trwającego 10 ms (1 cykl) nie można byłoby zaobserwować.

ZESPÓŁ TRZECH ZBIORNIKÓW

1. Budowa zespołu

Instalacja składa się z trzech zbiorników pokazanych na rysunku, z których dolny pełni rolę reaktora chemicznego. Zawory Z1, Z2 na dopływach do górnych zbiorników mają zapewnić, aby w zbiornikach tych nie zabrakło ciecży. Zawory Z3, Z4 i Z5 przeznaczone są dla reaktora, w którym ponadto znajduje się grzejnik GR i mieszadło M.




2. Sterowanie

- Zbiornik I – lewy górny

$h < X2$: nalewanie z zaworu Z1 do poziomu X2 w cyklu 3+1 s


$h \geq X2$: zamknięcie Z1 i oczekiwanie na opróżnienie zbiornika do poziomu X1 (pobór do reaktora).



- Zbiornik II – prawy górny

$h < X3$: zawór Z2 otwarty


$h \geq X3$: zawór Z2 otwarty do osiągnięcia poziomu X4, ale nie dłużej niż 10 s; potem zawór jest zamykany i oczekiwanie na opróżnienie zbiornika do poziomu X3.



- Zbiornik III – reaktor

$h < X5$: nalewanie z zaworów Z3, Z4 do poziomu X5, ale tylko wtedy, gdy w górnych zbiornikach poziomy przekraczają X1 i X3


$h \geq X5$: nalewanie z Z3, Z4 do poziomu X7 w cyklu 2 + 2 s



- $h \geq X7$: zamknięcie Z3, Z4, włączenie grzejnika GR aż do osiągnięcia temperatury TM, ale nie krócej niż 10 s i nie dłużej niż 20 s
- TM: wyłączenie GR, włączenie mieszadła M na 5 s
- opróżnianie: otwarcie zaworu Z5 aż poziom spadnie do X5, potem powrót do nalewania.

3. Grafy i fragmenty kodu

- Zbiornik I




```

STEROWNIK (PRG-ST)
0001 PROGRAM STEROWNIK
0002
0003 VAR_EXTERNAL X1,X2,X3,TM,Z1,Z2,Z3,GR,M: BOOL; END_VAR
0004
0005 VAR stan1: INT:=1; stan2: INT:=1; stan3: INT:=1;
0006     TON1,TON2,TON3,TON4,TON5,TON6,TON7:TON;
0007     _0:BOOL:=FALSE; _1:BOOL:=TRUE;
0008 END_VAR
0009
0001 (* Sterowanie zbiornikiem I *)
0002 TON1(IN:=stan1=1, PT:=T#3s);
0003 TON2(IN:=stan1=2, PT:=T#1s);
0004
0005 CASE stan1 OF
0006 1: Z1:=_1;
0007   IF NOT X2 AND TON1.Q THEN stan1:=2;
0008   ELSIF X2 THEN stan1:=3; END_IF
0009 2: Z1:=_0;
0010   IF NOT X2 AND TON2.Q THEN stan1:=1;
0011   ELSIF X2 THEN stan1:=3; END_IF
0012 3: Z1:=_0;
0013   IF NOT X1 THEN stan1:=1; END_IF
0014 END_CASE

```

Uwaga. Stanami automatów sterujących zbiornikami są stan 1, stan 2 i stan 3.

- Zbiornik II




```

0015
0016 (* Sterowanie zbiornikiem II *)
0017 TON3(IN:=stan2=2, PT:=T#10s);
0018
0019 CASE stan2 OF
0020 1: Z2:=_1;
0021 IF X3 THEN stan2:=2; END_IF
0022 2: Z2:=_1;
0023 IF X4 OR TON3.Q THEN stan2:=3; END_IF
0024 3: Z2:=_0;
0025 IF NOT X3 THEN stan2:=1; END_IF
0026 END_CASE
0027

```

- Zbiornik III



Wyjaśnienie. Otwarcie zaworów Z3, Z4 w stanie 1 następuje dopiero wtedy, gdy zbiorniki górne są przynajmniej minimalnie wypełnione.

```

0028 (* Sterowanie reaktorem *)
0029 TON4(IN:=stan3=2, PT:=T#2s);
0030 TON5(IN:=stan3=3, PT:=T#2s);
0031 TON6(IN:=stan3=4, PT:=T#20s);
0032 TON7(IN:=stan3=5, PT:=T#5s);
0033
0034 CASE stan3 OF
0035 1: Z5:=_0; GR:=_0; M:=_0;
0036 IF NOT X1 AND NOT X3 THEN Z3:=_0; Z4:=_0;
0037 ELSE Z3:=_1; Z4:=_0; END_IF
0038 IF X5 THEN stan3:=2; END_IF
0039 2: Z3:=_1; Z4:=_0; Z5:=_0; GR:=_0; M:=_0;
0040 IF NOT X7 AND TON4.Q THEN stan3:=3;
0041 ELSIF X7 THEN stan3:=4; END_IF
0042 3: Z3:=_0; Z4:=_1; Z5:=_0; GR:=_0; M:=_0;
0043 IF NOT X7 AND TON5.Q THEN stan3:=2;
0044 ELSIF X7 THEN stan3:=4; END_IF
0045 4: Z3:=_0; Z4:=_0; Z5:=_0; GR:=_1; M:=_0;
0046 IF TM AND (TON6.ET>T#10s) OR TON6.Q
0047 THEN stan3:=5; END_IF
0048 5: Z3:=_0; Z4:=_0; Z5:=_0; GR:=_0; M:=_1;
0049 IF TON7.Q THEN stan3:=6; END_IF
0050 6: Z3:=_0; Z4:=_0; Z5:=_1; GR:=_0; M:=_0;
0051 IF NOT X5 THEN stan3:=1; END_IF
0052 END_CASE
0053
0054

```

4. Symulacja

Zbiornik I

```
ZBIORNIK_I (PRG-ST)
0001 PROGRAM ZBIORNIK_I
0002 VAR_EXTERNAL
0003   X1,X2,Z1,Z3: BOOL;
0004 END_VAR
0005 VAR
0006   H:REAL;
0007 END_VAR
0008
0009
0001 IF Z1 THEN H:=H+0.1; END_IF
0002 IF Z3 THEN H:=H+0.05; END_IF
0003
0004 IF H<0.0 THEN H:=0.0; END_IF
0005 IF H>100.0 THEN H:=100.0; END_IF
0006
0007 IF H>=80.0 THEN X2:=TRUE;
0008     ELSE X2:=FALSE; END_IF
0009 IF H>=20.0 THEN X1:=TRUE;
0010     ELSE X1:=FALSE; END_IF
```

Zbiornik II

```
ZBIORNIK_II (PRG-ST)
0001 PROGRAM ZBIORNIK_II
0002 VAR_EXTERNAL
0003   X3,X4,Z2,Z4: BOOL;
0004 END_VAR
0005 VAR
0006   H:REAL;
0007 END_VAR
0008
0009
0001 IF Z2 THEN H:=H+0.1; END_IF
0002 IF Z4 THEN H:=H+0.05; END_IF
0003
0004 IF H<0.0 THEN H:=0.0; END_IF
0005 IF H>100.0 THEN H:=100.0; END_IF
0006
0007 IF H>=80.0 THEN X4:=TRUE;
0008     ELSE X4:=FALSE; END_IF
0009 IF H>=20.0 THEN X3:=TRUE;
0010     ELSE X3:=FALSE; END_IF
```

Zbiornik III

```
REAKTOR (PRG-ST)
0001 PROGRAM REAKTOR
0002 VAR_EXTERNAL
0003   X5,X6,X7,TM,Z3,Z4,Z5,GR: BOOL;
0004 END_VAR
0005 VAR
0006   H:REAL; TEMP:REAL:=20.0;
0007 END_VAR
0008
0009
0001 IF Z3 THEN H:=H+0.05; END_IF
0002 IF Z4 THEN H:=H+0.05; END_IF
0003 IF Z5 THEN H:=H+0.05; END_IF
0004
0005 IF H<0.0 THEN H:=0.0; END_IF
0006 IF H>100.0 THEN H:=100.0; END_IF
0007
0008 IF H>=80.0 THEN X7:=TRUE;
0009     ELSE X7:=FALSE; END_IF
0010 IF H>=50.0 THEN X6:=TRUE;
0011     ELSE X6:=FALSE; END_IF
0012 IF H>=20.0 THEN X5:=TRUE;
0013     ELSE X5:=FALSE; END_IF
0014
0015 IF Z3 OR Z4 THEN TEMP:=20.0; END_IF
0016 IF GR THEN TEMP:=TEMP+0.06; END_IF
0017
0018 IF TEMP>=80.0 THEN TM:=TRUE;
0019     ELSE TM:=FALSE; END_IF
```

Wyjaśnienia

1. Wartości 0.05/0.1 w równaniu dla H oznaczają procentową zmianę H na cykl obliczeń wynoszący 10 ms = 0.01 s. W przypadku 0.05 zmiana poziomu o 100% nastąpi w czasie

$$100\% \frac{0.01s}{0.05\%} = 20s,$$

a w przypadku 0.1 w czasie 10 s. Dopływ przez zawory Z1, Z2 do górnych zbiorników wynosi 0.1 i jest większy od poboru przez zawory Z3, Z4, którym jest 0.05.

2. Wartość 0.06 w równaniu dla TEMP oznacza zmianę TEMP w °C na cykl. Zmiana o 60°C nastąpi w czasie

$$60^{\circ}C \frac{0.01s}{0.6^{\circ}C} = 10s.$$

5. Program główny i zmienne globalne

```
MAIN (PRG-ST)
0001 PROGRAM MAIN
0002 VAR
0003 END_VAR
0004
0001 ZBIORNIK_I();
0002 ZBIORNIK_II();
0003 REAKTOR();
0004 STEROWNIK();
0005
```

```
Global_Variables
0001 VAR_GLOBAL
0002 X1,X2,X3,X4,X5,X6,X7,TM,Z1,Z2,Z3,Z4,Z5,GR,M: BOOL;
0003 END_VAR
0004
```

6. Run

